/*
 * Flashlib - QB9 common Flash library
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *     http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */


package com.qb9.flashlib.tasks
{
	/**
	 * A task container that updates subtasks only after a certain delay has passed.
	 *
	 * <p>Normally, tasks registered to a TaskRunner instance receive an update
	 * event whenever an ENTER_FRAME event is dispatched.  Use a Ticker when
	 * you need to run logic that runs on a fixed interval.  All subtasks will
	 * always be called with the given update interval.
	 * </p>
	 *
	 * @see TaskRunner
	 */
	public class Ticker extends TaskContainer
	{
		public var delay:uint;
		public var tickCount:uint;
		protected var accumulator:uint;

		/**
		 * Create a Ticker instance.
		 *
		 * @param delay Interval delay, in milliseconds.
		 */
		public function Ticker(delay:uint)
		{
			this.delay = delay;
			this.tickCount = 0;
			this.accumulator = 0;
		}
		
		/** @inheritDoc */
		public override function clone():ITask
		{
			var t:Ticker = new Ticker(delay);
			t.subtasks = cloneSubtasks();
			return t;
		}

		/** @inheritDoc */
		public override function update(milliseconds:uint):void
		{
			accumulator += milliseconds;
			while (accumulator >= delay)
			{
				tickCount++;
				accumulator -= delay;
				super.update(delay);
			}
		}
	}
}

